

PREGNANCY HISTORY MODEL ANALYSIS PLAN: DETERMINANTS OF STILLBIRTH

DHS METHODOLOGICAL REPORTS 34

AUGUST 2023

This publication was produced for review by the United States Agency for International Development. It was prepared by Sara Riese.

DHS Methodological Reports No. 34

Pregnancy History Model Analysis Plan: Determinants of Stillbirth

Sara Riese^{1,2}

ICF Rockville, Maryland, USA

August 2023

¹ ICF ² The DHS Program

Corresponding author: Sara Riese, International Health and Development, ICF, 530 Gaither Road, Suite 500, Rockville, MD 20850, USA; phone: 301-407-6500; fax: 301-407-6501; email: sara.riese@icf.com

Acknowledgments: The author wishes to thank Dr. Emma Chikovore for her thoughtful review and comments, and Trevor Croft for guidance on the new GR and NR datasets.

Editor: Diane Stoy Document Production: Chris Gramer

This study was carried out with support provided by the United States Agency for International Development (USAID) through The DHS Program (#720-OAA-18C-00083). The views expressed are those of the authors and do not necessarily reflect the views of USAID or the United States Government.

The DHS Program assists countries worldwide in the collection and use of data to monitor and evaluate population, health, and nutrition programs. Additional information about The DHS Program can be obtained from ICF, 530 Gaither Road, Suite 500, Rockville, MD 20850 USA; telephone: +1 301-407-6500, fax: +1 301-407-6501, email: info@DHSprogram.com, internet: www.DHSprogram.com.

Recommended citation:

Riese, Sara. 2023. *Pregnancy History Model Analysis Plan: Determinants Of Stillbirth*. DHS Methodological Reports No. 34. Rockville, Maryland, USA: ICF.

CONTENTS

	S			V /ii		
ACRON	NYMS A	ND ABBREVIAT	10NS	ix		
1	INTRO			1		
2	REQUI	REMENTS		3		
3	SETTI	G UP THE DAT	Α	5		
	3.1	Description of R	elevant Data Files and Variables in DHS-8	5		
4	MODE	ANALYSIS		7		
	4.1	Description of A	nalysis	7		
	4.2	Application and	Significance	7		
	4.3	Background The	eory	7		
	4.4	Analytic Details	· · ·	8		
		4.4.1 Creating	g the outcome variable	8		
		4.4.2 Calcula	tion of stillbirth rate	8		
		4.4.3 Coding	of covariates	8		
		4.4.4 Analytic	approach	9		
	4.5	Analytic Details.		9		
	4.6	Conclusions	······	3		
REFER	ENCES			5		
APPEN	DICES			17		
	Append	ix 1: Stata Do fil	e for stillbirth model analysis using GR data file	7		
	Appendix 2: Stata Do file for stillbirth model analysis using NR data file					

TABLES

6 7
7
/
10
11
12
13
•

PREFACE

The Demographic and Health Surveys (DHS) Program is one of the principal sources of international data on fertility, family planning, maternal and child health, nutrition, mortality, environmental health, HIV/AIDS, malaria, and provision of health services.

One of the objectives of The DHS Program is to continually assess and improve the methodology and procedures used to carry out national-level surveys as well as to offer additional tools for analysis. Improvements in methods used will enhance the accuracy and depth of information collected by The DHS Program and relied on by policymakers and program managers in low- and middle-income countries.

While data quality is a main topic of the DHS Methodological Reports series, the reports also examine issues of sampling, questionnaire comparability, survey procedures, and methodological approaches. The topics explored in this series are selected by The DHS Program in consultation with the U.S. Agency for International Development.

It is hoped that the DHS Methodological Reports will be useful to researchers, policymakers, and survey specialists, particularly those engaged in work in low- and middle-income countries, and will be used to enhance the quality and analysis of survey data.

Sunita Kishor Director, The DHS Program

ACRONYMS AND ABBREVIATIONS

ANC	antenatal	care

- DHS Demographic and Health Survey
- LMIC lower and middle-income country
- UNIGME United Nations Inter-agency Group for Child Mortality Estimation

1 INTRODUCTION

This document is a model analysis plan for the further analysis of data from the Demographic and Health Surveys (DHS) Program. This plan is designed for use by students, researchers, and other data users for developing country-specific further analysis plans. The plan has been released to coincide with the release of data from the eighth round of the DHS, which switched from using a birth history to pregnancy history as the basis for the reproductive calendar part of the Woman's Questionnaire.

The decision to switch to pregnancy history was made primarily to enable better capture of information on stillbirth. Birth histories may miss some stillbirths and early neonatal deaths.^{1–3} Recent evidence suggests that pregnancy history is associated with increased reporting of births, including stillbirths, and early neonatal deaths compared to birth history.⁴ This model analysis therefore focuses on stillbirth—measuring the stillbirth rate and identifying determinants of stillbirth.

This model analysis plan is organized to show researchers each step of the further analysis process.

REQUIREMENTS

Stata data analysis software is required because this model analysis plan is designed for users of Stata. In addition, the DHS datasets are required for this analysis. For most introductory training or model purposes, the DHS Program recommends using the model datasets, which have been created for practice and do not represent any country's actual data. However, the model datasets are based on the DHS-6 questionnaire and recode dataset. The datasets do not include the full pregnancy history and cannot be used for this model analysis plan. This analysis can only be run with a DHS-8 or later recode dataset. After registering, users can download the relevant data files from The DHS Program website.

3 SETTING UP THE DATA

3.1 Description of Relevant Data Files and Variables in DHS-8

With the introduction of the pregnancy history in DHS8, new data files were created to include information on each pregnancy. The key datasets in this analysis are the GR file and the NR file. In previous DHS dataset releases, information on women was found in the IR file, and information on births in the BR file. While the IR and BR files are still available, pregnancy information is now located in the GR and NR files.

The GR file in DHS8 is the new dataset that contains information on all pregnancies of interviewed women. In the GR file, the pregnancy is the unit of analysis, compared to the previous IR file, where the interviewed woman was the unit of analysis. In the GR file most of the variable numbers are the same as the variable numbers in the IR and BR file. New variables related to each pregnancy have been added, and these variables begin with the letter P. Some of these new variables link the pregnancy history to the birth history. For example, PIDXB is the index from the pregnancy history to the birth history. PIDXB is blank if the pregnancy outcome is not a live birth. Some of these new variables provide information about the pregnancy and its outcome. PIDX is the pregnancy history index variable, while PORD is the pregnancy order. The duration of pregnancy is given in p20 and p21, the pregnancy outcome declared by the respondent is given in p30, and the pregnancy outcome is reclassified in p32.

Pregnancy is also the unit of analysis in the NR file, but these pregnancies are limited to those in the previous three years. In the NR file, the variable names are the same as in the KR file. The NR file includes the pregnancy and postnatal care variables from previous DHS phases, such as number of antenatal care (ANC) visits, receipt of specific aspects of ANC, and location of delivery. While the NR file includes all the variables cited in the previous paragraph, there are also additional variables that link the pregnancy to the birth and pregnancy history. Where these variables started with P in the GR file, they start with M in the NR File: MIDX is the index to the birth history and is left blank for non-live births, while MIDXP is the index to the pregnancy history. The variable m80 categorizes pregnancy outcomes from pregnancies in the previous three years as either a live birth, a stillbirth, or a miscarriage/abortion.

Table 1 summarizes these new pregnancy history variables in the GR and NR datasets.

Variable name	Definition and examples
All pregnancies	
PIDX	Pregnancy history index variable
	1 = most recent pregnancy 2 = pregnancy just before the most recent pregnancy 3 = pregnancy before pregnancy with PIDX=2
	Continues for all pregnancies
PIDXB	Index variable from the pregnancy history to the birth history
	Lists the number where the pregnancy can be found in the birth history. Equal to the value of PIDX ONLY IF all pregnancies ended in a live birth 0 = termination (not in birth history)
PORD	Pregnancy order variable
	1 = first pregnancy 2 = second pregnancy 3 = third pregnancy 4 = fourth pregnancy
	Continues for all pregnancies
p20	Duration of pregnancy in months (1–10)
p21	Duration of pregnancy in days
p30	Pregnancy outcome declared by respondent
	1 = born alive 2 = born dead 3 = miscarriage 4 = abortion
p32	Pregnancy outcome reclassified
	1 = born alive 2 = born dead 3 = miscarriage 4 = abortion
	Pregnancy outcome may be reclassified if, for example, a respondent says she had a stillbirth at 5 months gestation. This outcome would be reclassified as a miscarriage.
Pregnancies in the la	st 3 years (included in the GR file, but missing for all pregnancies from earlier than 3 years ago)
m80	Pregnancy outcome variable
	1 = most recent live birth 2 = prior live birth 3 = most recent stillbirth 4 = prior stillbirth 5 = miscarriage/abortion
MIDX	Index from the pregnancies in the past three years to the birth history
	Lists the number where the pregnancy can be found in the birth history. Equal to the value of PIDX ONLY IF all pregnancies ended in a live birth 0 = it is not a birth
MIDXP	Pregnancy history index variable for pregnancies in the past 3 years
	1 = most recent pregnancy 2 = pregnancy just before the most recent pregnancy
	Continues for all pregnancies in the past 3 years

Table 1 Summary of select new key pregnancy history variables

In this analysis, we will show analyses using both the GR and NR datasets.

4 MODEL ANALYSIS

4.1 Description of Analysis

This analysis estimates the prevalence of stillbirth in a population, and then seeks to identify factors that are associated with stillbirth in that population.

4.2 Application and Significance

Findings from analyses such as these can be used by program managers and policy makers to identify the factors that contribute to stillbirth so that these factors can be prioritized for interventions that can decrease the stillbirth rate.

4.3 Background Theory

Before moving to the data analysis, it is important to develop a logical framework for the research question. In this case, we look to the previous literature to identify factors that influence the risk of stillbirth.

Table 2 categorizes the recent evidence^{5,6} on maternal, pregnancy, birth, or fetal characteristics that are associated with stillbirth:

Maternal characteristics	Pregnancy characteristics	Birth characteristics	Fetal characteristics
Stillbirth history in previous pregnancies	Primipara	Non-medically assisted birth	Low birth weight (<2,500g)
Maternal infection	Multiple pregnancy	Post-term birth	Male
Age over 31	Lack of antenatal care		
Maternal chronic disease (hypertension, diabetes)			
Tobacco and alcohol use			

Table 2 Characteristics associated with stillbirth

All of these variables may not be available in the dataset. While most DHS datasets have a standard set of variables, each country can include country-specific questions. Thus, it is important to review the specific dataset you are working with before proceeding with the analysis. We generally do not have data on maternal chronic disease or tobacco and alcohol use on all mothers. In addition, perceived size at birth and sex are available only for pregnancies that ended in a live birth.

With the pregnancy history, it is important to consider the inclusion of time-varying characteristics. Many sociodemographic characteristics of women, such as education, employment status, wealth quintile, and residence are collected only at the time of the interview. Some characteristics may have been the same at the time of the stillbirth, although some may have changed. For example, if a woman had a pregnancy that ended in stillbirth when she was young, her level of education may have been lower at that time. However, since that time, the woman may have completed more education. Including some of these time-varying characteristics in your model may be appropriate, although it is important to discuss this in your limitations section.

4.4 Analytic Details

This analysis is presented with the GR file and the NR File. Code for these analyses is available in the appendix of this report and at The DHS Program's GitHub page. It is important to consider your specific research questions when determining which dataset to use. Since stillbirth is a rare event, the sample size using one country's NR file, which includes only pregnancies from the past three years, is likely to be small. Using the GR file, which includes all pregnancies, allows for a larger sample size. If your study is multicountry, then pooling the datasets would include the ANC variables available in the NR file in your analysis. The GR file includes information on all pregnancies in a woman's reproductive life. With some outcomes, there may be concern about recall bias when using a woman's entire reproductive lifespan as a time period for the outcome. However, stillbirth is a memorable event that women are unlikely to forget. Another concern is that some characteristics of a woman, which were collected at the time of the survey, may not apply to her at the time she had the stillbirth. This may be especially true if the stillbirth occurred far in the past. In addition, some variables about specific pregnancies, such as if the woman had ANC and from whom she received such care, are included only for the most recent pregnancies (NR file).

4.4.1 Creating the outcome variable

The International Classification of Diseases 11th Revision identifies stillbirth as a baby born with no signs of life at 22 or more completed weeks of gestation.⁷ Stillbirths are categorized as early gestation stillbirth if they occur at 22–27 completed weeks of gestation, or late gestation stillbirths if they occur at 28 or more completed weeks of gestation. The United Nations Inter-agency Group for Child Mortality Estimation (UNIGME) and The DHS Program use estimates of late gestational stillbirth.^{8,9}

Therefore, the outcome variable of stillbirth should be limited to all pregnancies that progress to at least 28 completed weeks (7 completed months) of gestation. In the DHS datasets, since information on pregnancies is captured in completed months and not weeks, we use the cut-off of less than 8 completed months. Since the GR and NR files include all pregnancies, including those that ended before 28 weeks, we dropped all cases where p20 < 8.

4.4.2 Calculation of stillbirth rate

The stillbirth rate is reported per 1,000 births from 28 weeks or later. The DHS Program calculates the stillbirth rate in the 5-year period before the survey. Pregnancies that progressed past 7 months before this period should *not* be included in the calculation of the stillbirth rate. The stillbirth rate in the 5-year period before the survey is a commonly reported indicator included in the DHS Final Report. Thus, you do not need to include "estimation of the stillbirth rate" as a research question or as a part of the results. Instead, it can be presented in the background or introduction section of your paper.

4.4.3 Coding of covariates

Many variables in this analysis do not need to be recoded from the DHS8 GR recode file, although some need to be created in your do file before being used in the analysis. These include age at stillbirth, history of stillbirth, and history of child death. For guidance on how to categorize any variables or indicators to match The DHS Program, please refer to DHS indicators section of The DHS Program's GitHub page.

4.4.4 Analytic approach

The most common approach to identifying determinants of a bivariate outcome is logistic regression. The analytic approach described here follows the standard steps of 1) describing the population, 2) describing the bivariate associations of the covariates with the outcome, and 3) describing the multivariate associations of the outcome.

After you have coded your outcomes and covariates, it is important to compare your estimates to the findings in the DHS Final Report from the same survey to ensure that the coding has been done correctly. Any variables that are reported in the final report with the same categorizations should be checked to confirm that the coding is correct.

All covariates in your logical framework that are also available should be included in your regression models. However, in some cases, you may have covariates that are highly correlated. Logistic regression assumes that there is little to no multicollinearity between the independent variables. Thus, correlation between the independent variables needs to be checked before running the regression. For example, woman's age and parity are frequently highly correlated. In this case, it is recommended that you choose only one of those variables to retain in your analysis. In addition, some variables might have large amounts of missing data because of skip patterns in the data. These variables may need to be omitted from the multivariable analysis because they reduce the overall sample size available for the analysis.

4.5 Analytic Details

Table 3a shows the distribution of covariates for all reported pregnancies of interviewed women lasting at least 7 months as well as the crosstabulation of those covariates by the outcome of interest—whether or not the pregnancy ended in a stillbirth in Country A. A higher proportion of pregnancies ending in stillbirth were among women in the oldest age group at birth compared to pregnancies ending in a live birth (8.5% compared with 5.4%). There are significant differences in the level of education, current employment, smoking status, history of child death, multiple pregnancy, and duration of pregnancy when comparing pregnancies that ended in a stillbirth and those that ended in a live birth.

	Total	Live Birth	Stillbirth	p value
	N = 78,423	N = 77,299	N = 1,124	
Total		95.6%	1.4%	
Maternal characteristics	%	%	%	
Residence				.14
Urban	30.9	30.9	32.9	
	09.1	09.1	07.1	05
Lowest	31.0	31.0	33.4	.25
Second	19.6	19.6	18.1	
Middle	19.4	19.4	17.8	
Fourth	18.0	18.0	18.1	
Highest	12.0	12.0	12.6	
Maternal age at birth				<.001
12–25	60.2	60.2	55.6	
26-35 26 50	34.4	34.4	35.9	
30-30	5.5	5.4	0.0	
Current marital status	4.0	4.2	2.2	.13
Ever married	4.2	4.2	3.3	
	33.0	33.0	50.7	
Education	60.7	60.6	70 /	.006
Secondary or higher	09.7 30.3	09.0 30.4	73.4	
	50.5		20.0	000
Employment status	27.7	27 7	11 E	.008
Employed or employed at some	51.1	57.7	41.5	
point in the last 12 months	62.3	62.3	58.5	
Smokor				024
Yes	0.4	0.4	0.8	.024
No	52.2	52.2	55.1	
Missing	47.4	47.4	44.1	
History of stillbirth				< 001
Yes	2.6	2.6	0.0	
No	97.4	97.4	100.0	
History of child death				<.001
Yes	11.0	11.0	14.6	
No	89.0	89.0	85.4	
Pregnancy characteristics				
First birth				.064
Yes	28.8	28.8	31.3	
No	71.2	71.2	68.7	
Multiple pregnancy				<.001
Yes	2.9	2.9	7.5	
No	97.1	97.1	92.5	
Duration of pregnancy				<.001
7 months	1.4	1.1	22.4	
8 months	5.5	5.3	20.0	
9 months	91.0	91.5	54.2 3 4	
	2.2	2.1	5.4	

Table 3aDistribution and crosstabulation of pregnancies ending in stillbirth, by maternal and
pregnancy characteristics using the GR file: Country A 2023 DHS

Table 3b shows the same distribution and crosstabulations, but with the NR dataset, it is limited to pregnancies in the previous three years in Country A. As expected, the sample size is much smaller. Many of the same covariates are significant, although in this population, the *p* values of employment status and current smoking status are now over .05 and are no longer significant. In the NR file, we also have additional variables on characteristics of the pregnancy and birth, which include variables on ANC and location of the birth. A higher proportion of pregnancies that ended in stillbirth had non-skilled assistance for ANC or did not attend ANC.

	Total	Live Birth	Stillbirth	p value
	N = 11.916	N = 11.709	N = 207	produce
Total	,	98.3%	1.7%	
Maternal characteristics	%	%	%	
Residence				.10
Urban	34.2	34.2	39.6	
Rural	65.8	65.8	60.4	
Household wealth quintile				.56
Lowest	32.7	32.8	31.9	
Middle	17.4	17.4	17.9	
Fourth	19.0	18.9	22.7	
Highest	14.0	14.0	14.0	
Maternal age at birth				.019
12-25	47.4	47.6	40.1	
36–50	10.7	41.8	44.0 15.9	
Current marital status				15
Never in union	9.7	9.7	6.8	.10
Ever married	90.3	90.3	93.2	
Education				.036
None or primary education	55.6	55.5	62.8	
Secondary or higher	44.4	44.5	37.2	
Employment status	50.0	50.4	50.0	.67
Final Function of the second s	52.2	52.1	53.6	
point in the last 12 months	47.8	47.9	46.4	
Currently smokes cigarettes				.072
Yes	0.3	0.3	0.5	
No	52.2	52.1	59.9	
Missing	47.4	47.0	39.6	
History of stillbirth	0.4	0.4	0.0	.34
No	99.6	99.6	100.0	
History of child death				041
Yes	1.0	1.0	2.4	.0+1
No	99.0	99.0	97.6	
Pregnancy characteristics	%	%	%	
First birth				.87
Yes	25.1	25.2	24.6	
No	74.9	74.8	75.4	
Multiple pregnancy				<.001
Yes	3.2	3.2	8.7 01 3	
	50.0	50.0	51.5	. 004
Skilled assistance during ANC Skilled provider	85.3	85.4	82.1	<.001
Other/no one	3.6	3.4	11.6	
Missing	11.1	11.2	6.3	
Birth characteristics	%	%	%	
Live births by place of delivery				.042
Home/other	24.9	24.9	26.1	
Health facility	75.1	75.1	73.9	
Duration of pregnancy	4.0	4.0	05.4	<.001
7 months 8 months	1.8 6.1	1.3	25.1 20.3	
9 months	89.6	90.3	48.8	
10 months	2.5	2.5	5.8	

Table 3bDistribution and crosstabulation of pregnancies ending in stillbirth, by maternal,
pregnancy, and birth characteristics using the NR file: Country A 2023 DHS

Before performing the regression analysis, we dropped current smoking status from the model due to large amounts of missing data. We considered dropping maternal age due to correlation with first birth, but the sensitivity analysis showed that the medium correlation (.3 in the population of all pregnancies, .45 in the population of recent pregnancies) did not affect the model results. Therefore, both variables were retained.

Bivariate and multivariate logistic regression results are presented in Table 4a and 4b. In the population of all pregnancies, the odds of stillbirth are higher for women at higher age at birth and for those with a history of child death. The odds are higher for multiple pregnancies and lower for the longer the duration of pregnancy. In the multivariable regression, these significant associations are maintained, and the first birth is also significantly associated with stillbirth.

	N = 78,423		N = 7	6,419
	OR	95% CI	aOR	95% CI
Maternal characteristics				
Residence (Ref: Urban)	0.89	[0.72, 1.10]	0.89	[0.63,1.27]
Household wealth quintile (Ref: Lowest)				
Second	0.84	[0.67,1.06]	1.11	[0.79, 1.55]
Middle	0.86	[0.66, 1.13]	1.12	[0.79, 1.60]
Fourth	1.03	[0.79, 1.33]	0.97	[0.62, 1.51]
Highest	0.93	[0.68, 1.29]	0.88	[0.52, 1.50]
Maternal age at birth (Ref: 12–25)				
26–35	1.17	[0.99, 1.39]	1.38**	[1.12, 1.69]
36–50	1.67***	[1.26, 2.20]	1.86***	[1.34, 2.60]
Current marital status (Ref: Never in union)				
Ever married	1.37	[0.90, 2.10]	1.17	[0.67, 2.05]
Education (Ref: None or primary education)				
Secondary or higher	0.90	[0.74, 1.09]	0.81	[0.61, 1.07]
Employment Status (Ref: Not employed)				
Employed	0.88	[0.74, 1.06]	1.00	[0.78, 1.28]
Smoker (Ref: Not a smoker) N = 41,285	1.29	[0.58, 2.85]	-	-
History of stillbirth (Ref: No history of stillbirth)				
History of child death (Ref: No history of child death)	1.38**	[1.08, 1.76]	1.51*	[1.07, 2.13]
Pregnancy characteristics				
First birth (Ref: Not first birth)	1.09	[0.93, 1.28]	1.42***	[1.16, 1.74]
Multiple pregnancy (Ref: Single pregnancy)	2.77***	[1.96, 3.93]	2.35***	[1.45, 3.79]
Duration of pregnancy (Ref: 7 months)				
8 months	0.15***	[0.11, 0.20]	0.14***	[0.11, 0.19]
9 months	0.03***	[0.02, 0.04]	0.03***	[0.02, 0.04]
10 months	0.06***	[0.04, 0.10]	0.06***	[0.04, 0.10]

Table 4aBivariate and multivariable logistic regression results using the GR file: Country A 2023
DHS

In the population of pregnancies within the last three years, there are similar, but not identical associations. In the bivariate logistic regression shown in Table 4b, being in the oldest age group at birth compared to being in the youngest, having a multiple pregnancy compared to a single pregnancy, and receiving ANC care from a non-skilled provider or no one compared to a skilled provider were associated with increased odds of stillbirth. Longer duration of pregnancy was associated with lower odds of stillbirth. In the multivariable model, all covariates maintained their significance, except for maternal age, which became non-significant after controlling for all covariates.

Table 4b	Bivariate and multivariable logistic regression results using the NR file: Country A 2023
	DHS

N = 11,916		N = 10,590	
OR	95% CI	aOR	95% CI
0.72	[0.46, 1.13]	0.73	[0.44, 1.23]
0.73 1.23 1.22 1.09	[0.41, 1.30] [0.68, 2.22] [0.74, 2.01] [0.53, 2.27]	0.79 1.09 1.03 0.94	[0.43, 1.46] [0.60, 1.99] [0.54, 1.96] [0.44, 1.99]
1.43 1.97*	[0.88, 2.33] [1.12, 3.47]	1.40 1.77	[0.81, 2.42] [0.91, 3.44]
1.71	[0.86, 3.42]	1.45	[0.72, 2.92]
0.82	[0.53, 1.27]	0.93	[0.59, 1.45]
1.00	[0.65, 1.55]	1.00	[0.67, 1.50]
1.53	[0.22, 10.66]	-	-
2.17	[0.77, 6.09]	1.99	[0.74, 5.29]
0.78	[0.52, 1.18]	1.10	[0.61, 2.01]
3.37***	[1.67, 6.78]	4.18***	[2.06, 8.47]
0.15*** 0.04*** 0.10***	[0.08, 0.27] [0.02, 0.06] [0.04, 0.23]	0.16*** 0.03*** 0.10***	[0.09, 0.30] [0.02, 0.07] [0.04, 0.25]
4.35***	[2.43, 7.80]	4.77***	[2.74, 8.29]
1.09	[0 70 1 70]	1 20	[0.75, 1.93]
	N = OR 0.72 0.73 1.23 1.22 1.09 1.43 1.97* 1.71 0.82 1.00 1.53 2.17 0.78 3.37*** 0.15*** 0.04*** 0.10*** 4.35***	N = 11,916 OR 95% CI 0.72 $[0.46, 1.13]$ 0.73 $[0.41, 1.30]$ 1.23 $[0.68, 2.22]$ 1.22 $[0.74, 2.01]$ 1.09 $[0.53, 2.27]$ 1.43 $[0.88, 2.33]$ 1.97^* $[1.12, 3.47]$ 1.71 $[0.86, 3.42]$ 0.82 $[0.53, 1.27]$ 1.00 $[0.65, 1.55]$ 1.53 $[0.22, 10.66]$ 2.17 $[0.77, 6.09]$ 0.78 $[0.52, 1.18]$ 3.37^{***} $[1.67, 6.78]$ 0.15^{***} $[0.02, 0.06]$ 0.10^{***} $[0.02, 0.06]$ 0.10^{***} $[2.43, 7.80]$	N = 11,916 N = 1 OR 95% CI aOR 0.72 $[0.46, 1.13]$ 0.73 0.73 $[0.41, 1.30]$ 0.79 1.23 $[0.68, 2.22]$ 1.09 1.22 $[0.74, 2.01]$ 1.03 1.09 $[0.53, 2.27]$ 0.94 1.43 $[0.88, 2.33]$ 1.40 1.97^* $[1.12, 3.47]$ 1.77 1.71 $[0.86, 3.42]$ 1.45 0.82 $[0.53, 1.27]$ 0.93 1.00 $[0.65, 1.55]$ 1.00 1.53 $[0.22, 10.66]$ - 2.17 $[0.77, 6.09]$ 1.99 0.78 $[0.52, 1.18]$ 1.10 3.37^{***} $[1.67, 6.78]$ 4.18^{***} 0.15^{***} $[0.02, 0.06]$ 0.03^{***} 0.10^{***} $[0.4, 0.23]$ 0.10^{***} 4.35^{***} $[2.43, 7.80]$ 4.77^{***}

4.6 Conclusions

The full pregnancy history can be used for many new analyses of different pregnancy outcomes, including stillbirth. This model analysis presents one type of analysis aimed at identifying characteristics of women, pregnancies, and births that are associated with stillbirth. This type of analysis is relevant for many policy and programming decisions. In this model analysis, policies aimed at reducing stillbirth in Country A could focus on increasing ANC and increasing utilization of skilled assistance during ANC as a strategy, since women who had skilled assistance during ANC had lower odds of stillbirth. This analysis can be adapted to answer a variety of questions about pregnancy outcomes. Results from this type of analysis should be interpreted carefully. When using the pregnancy history and controlling for current characteristics, it is important to consider the length of time since the pregnancy and if the current characteristics are time-varying and may have changed from the time of the pregnancy to the time of data collection.

REFERENCES

- 1. Bradley SEK, Winfrey W, Croft TN. *Contraceptive Use and Perinatal Mortality in the DHS: An Assessment of the Quality and Consistency of Calendars and Histories.* 2015. DHS Methodological Reports No 17. http://dhsprogram.com/pubs/pdf/MR17/MR17.pdf
- 2. Espeut D, Becker S. The validity of birth and pregnancy histories in rural Bangladesh. *J Health, Popul Nutr.* 2015;33(1):1–11. https://doi.org/10.1186/s41043-015-0027-8
- 3. MacQuarrie KLD, Winfrey W, Meijer-Irons J, Morse AR. *Consistency of reporting of terminated pregnancies in DHS calendars*. 2018. *DHS Methodological Reports No 25*. http://dhsprogram.com/pubs/pdf/MR25/MR25.pdf
- 4. Akuze J, Blencowe H, Waiswa P, et al. Randomised comparison of two household survey modules for measuring stillbirths and neonatal deaths in five countries: the Every Newborn-Indepth Study. *Lancet Glob Health*. 2020;8(4):e555-e566. https://doi.org/10.1016/S2214-109X(20)30044-9
- Rambliere L, de Lauzanne A, Diouf JB, et al. Stillbirths and neonatal mortality in LMICs: A community-based mother-infant cohort study. *J Global Health*. 2023;1304031. https://jogh.org/2023/jogh-13-040316
- 6. Lawn JE, Blencowe H, Waiswa P, et al. Stillbirths: rates, risk factors, and acceleration towards 2030. *Lancet*. 387(10018):587–603. https://doi.org/10.1016/S0140-6736(15)00837-5
- 7. World Health Organziaion. *International Classification of Diseases Eleventh Revision (ICD-11)*. Geneva: World Health Organization; 2022. https://icd.who.int/en
- 8. Croft TN, Marshall, AMJ, Allen, CK, et al. *Guide to DHS Statistics*. 2018. Rockville: ICF. https://dhsprogram.com/data/Guide-to-DHS-Statistics/index.cfm
- United Nations Inter-agency Group for Child Mortality Estimation (UN IGME). Never Forgotten: The situation of stillbirth around the globe. 2023. New York: UNICEF. https://data.unicef.org/resources/never-forgotten-stillbirth-estimates-report/

APPENDICES

Appendix 1: Stata Do file for stillbirth model analysis using GR data file

Program: GR stillbirth.do Purpose: Stillbirth model analysis - determinants of stillbirth using the GR datafile Author: Sara Riese Date last modified: June 23, 2023 *ssc install fre //If you do not have the fre command installed, you will need to do so before running this code global grdata "KHGR81FL PHGR81FL NPGR81FL KEGR8AFL" //All available GR files as of July 2023 foreach c in \$grdata { *PART 1: LOAD DATASET FOR USE cd "yourdatapath" //set your directory to the folder where your data are located use "`c'.DTA", clear gen cn=substr("`c'",1,2) scalar cn=cn[1] local cn=cn label define YN 0 "No" 1 "Yes" *Drop pregnancies that ended before 7 months, and miscarriages/abortions. We are interested in stillbirth vs live births keep if p20>=7 & (p32==1 | p32==2) ***PART 2: CODING VARIABLES** ****** *Pregnancy ended in a stillbirth gen stillbirth=0 replace stillbirth=1 if p32==2 label values stillbirth YN *****Calculating the Stillbirth Rate****** preserve * Set length of history to use * Here we use the past 5 years (59 months) gen i=v008-p3+v018 gen callen = v018 + 59 gen beg = v018 gen end = callen

 \star Include only the five year period preceding the survey, and keep only births and stillbirths keep if i >= beg & i <= end & (p32 == 1 | p32 == 2)

* calculate the stillbirth rate using weights tab stillbirth [iw=v005/1000000] restore ***** ***** ****Available in the GR file (Full pregnancy history) *Urban/rural v025 recode v025 (1=0 "Urban") (2=1 "Rural"), gen(resid) label var resid "Residence" *Education level v106 recode v106 (0/1=0 "No/primary education") (2/3=1 "Secondary or Higher Education"), gen(edu) *Marital status v501 recode v501 (0=0 "Never in Union") (1/5=1 "Ever Married"), gen(maritalstat) *Employment status v731 //Employment status recode v731 (0=0 "Not employed in last 12 months") (1/3=1 "Employed or employed at some point in the last 12 months") (8=9 "Don't know/missing"), gen(rc_empl) label var rc_empl "Employment status" *Wealth index v190 *Smoker v463a* BE SURE TO LOOK AT THE DISTRIBUTION OF THIS QUESTION, THERE MAY BE LARGE % OF MISSING DATA DUE TO SKIP PATTERN *Age at birth p3-v011 gen age_b=p3-v011 fre age_b gen age_b_yr=int(age_b/12) fre age_b_yr recode age_b_yr (min/25=1 "15-25") (26/35=2 "26-35") (36/50=3 "36-49"), gen(age_b_yr_cat) *Primipara gen primi=0 replace primi=1 if pord==1 *History of stillbirth gen sbpreg=. replace sbpreg=pord if stillbirth==1 bysort caseid (sbpreg): replace sbpreg = sbpreg[1] if missing(sbpreg) gen histsb=0 if pord<=sbpreg</pre> replace histsb=1 if pord>sbpreg & sbpreg!=. *History of child death gen ch_death=0 replace ch_death=1 if p5==0 & p32==1 gen deathpreg=. replace deathpreg=pord if ch_death==1 bysort caseid (deathpreg): replace deathpreg = deathpreg[1] if missing(deathpreg) gen hist_chdeath=0 if pord<=deathpreg</pre> replace hist_chdeath=1 if pord>deathpreg & deathpreg!=.

```
*Multiple pregnancy
gen multpreg=0
replace multpreg=1 if p0>=1 & p0!=.
*Gestational age at birth p20
****Descriptive analysis
foreach var of varlist v025 edu maritalstat rc_empl v190 v463a age_b_yr_cat primi-multpreg p20 {
       tab `var', m
}
*Check for correlations between variables
pwcorr resid edu maritalstat rc_empl v190 v463a age_b_yr_cat primi histsb hist_chdeath multpreg
p20
*v463a dropped due to high level of missing data
*histsb dropped due to lack of cases
global stillbirthGRcovs "resid edu maritalstat rc_empl v190 age_b_yr_cat primi hist_chdeath
multpreg p20"
global istillbirthGRcovs "i.resid i.edu i.maritalstat i.rc_empl i.v190 i.age_b_yr_cat i.primi
i.hist_chdeath i.multpreg i.p20"
*** Set the complex survey design ***
gen wt=v005/1000000
svyset [pw=wt], psu(v021) strata(v022)
foreach x in $istillbirthGRcovs {
              svy:logistic stillbirth `x' //Bivariate associations
              outreg2 using "GR_biregs_`c'.xls", eform stats(coef ci) alpha(0.001, 0.01, 0.05)
sideway dec(2) label(insert) noparen nocons append
}
svy: logistic stillbirth i.resid i.edu i.maritalstat i.rc_empl i.v190 i.age_b_yr_cat i.primi
i.hist_chdeath i.multpreg i.p20
outreg2 using "GR_multiregs_`c'.xls", eform stats(coef ci) alpha(0.001, 0.01, 0.05) sideway
dec(2) label(insert) noparen nocons append
save "`cn'GR_coded.DTA", replace
}
```

Appendix 2: Stata Do file for stillbirth model analysis using NR data file

Program: NR stillbirth.do Purpose: Stillbirth model analysis - determinants of stillbirth using the NR datafile Author: Sara Riese Date last modified: June 23, 2023 *ssc install fre //If you do not have the fre command installed, you will need to do so before running this code Some health related variables of interest are only available in the NR file (all pregnancies in last 3 years) global nrdata "KHNR81FL PHNR81FL NPNR81FL KENR8AFL" //All availble NR files as of July 2023 foreach c in \$nrdata { *PART 1: LOAD DATASET FOR USE cd "yourdatapath" //set your directory to the folder where your data are located use "`c'.DTA", clear gen cn=substr("`c'",1,2) local cn=cn[1] *Drop pregnancies that ended before 7 months, and miscarriages/abortions. We are interested in stillbirth vs live births keep if p20>=7 & (p32==1 | p32==2) ***PART 2: CODING VARIABLES** ***** *Pregnancy ended in a stillbirth gen stillbirth=0 replace stillbirth=1 if p32==2 label define YN 0 "No" 1 "Yes" label values stillbirth YN *Urban/rural v025 recode v025 (1=0 "Urban") (2=1 "Rural"), gen(resid) label var resid "Residence" *Education level v106 recode v106 (0/1=0 "No/primary education") (2/3=1 "Secondary or Higher Education"), gen(edu) *Marital status v501 recode v501 (0=0 "Never in Union") (1/5=1 "Ever Married"), gen(maritalstat) *Employment status v731 //Employment status recode v731 (0=0 "Not employed in last 12 months") (1/3=1 "Employed or employed at some point in the last 12 months") (8=9 "Don't know/missing"), gen(rc_empl) label var rc_empl "Employment status"

```
*Wealth index v190
```

Smoker v463a BE SURE TO LOOK AT THE DISTRIBUTION OF THIS QUESTION, THERE MAY BE LARGE % OF MISSING DATA DUE TO SKIP PATTERN *Age at birth p3-v011 gen age_b=p3-v011 fre age_b gen age_b_yr=int(age_b/12) fre age_b_yr recode age_b_yr (min/25=1 "12-25") (26/35=2 "26-35") (36/50=3 "36-50"), gen(age_b_yr_cat) *Primipara (stillbirth) pord==1 gen primi=0 replace primi=1 if pord==1 *History of stillbirth gen sbpreg=. replace sbpreg=pord if stillbirth==1 bysort caseid (sbpreg): replace sbpreg = sbpreg[1] if missing(sbpreg) gen histsb=0 if pord<=sbpreg</pre> replace histsb=1 if pord>sbpreg & sbpreg!=. *History of child death gen ch_death=0 replace ch death=1 if p5==0 & p32==1 gen deathpreg=. replace deathpreg=pord if ch_death==1 bysort caseid (deathpreg): replace deathpreg = deathpreg[1] if missing(deathpreg) gen hist_chdeath=0 if pord<=deathpreg</pre> replace hist_chdeath=1 if pord>deathpreg & deathpreg!=. *Multiple pregnancy gen multpreg=0 replace multpreg=1 if p0>=1 & p0!=. *Gestational age at birth p20 *Number of ANC visits //ANC by type of provider ** Note: Please check the final report for this indicator to determine the categories and adjust the code and label accordingly. gen rh_anc_pv = 6 if m2a! = . = 4 if m2g == 1 | m2h == 1 | m2k == 1 replace rh_anc_pv replace rh_anc_pv = 3 if m2c == 1 replace rh_anc_pv replace rh_anc_pv = 2 if m2b == 1 = 1 if m2a == 1 replace rh_anc_pv = 9 if m2a == 9 label define rh_anc_pv /// 1 "Doctor" /// 2 "Nurse/midwife" /// 3 "Auxiliary midwife" /// 4 "TBA/CHW/other/relative" 111 9 "Missing" /// 6 "No ANC" label val rh_anc_pv rh_anc_pv

//ANC by skilled provider ** Note: For Cambodia 2022 FR, doctor, nurse, auxiliary midwife is considered skilled. recode rh_anc_pv (1/3 = 1 "Skilled provider") (4/6 = 2 "Other/No one") (9=3 "Don't know/missing"), gen(rh_anc_pvskill) *recode rh_anc_pv (1/3 = 1 "Skilled provider") (4/9 = 0 "Unskilled/no one") , gen(rh_anc_pvskill) label var rh_anc_pvskill "Skilled assistance during ANC" //Number of ANC visits in 4 categories that match the table in the final report recode m14 (0=0 "none") (1=1) (2 3=2 "2-3") (4/90=3 "4+") (else=9 "don't know/missing"), gen(rh_anc_numvs) label var rh_anc_numvs "Number of ANC visits" //4+ ANC visits recode rh_anc_numvs (1 2 9=0 "no") (3=1 "yes"), gen(rh_anc_4vs) lab var rh_anc_4vs "Attended 4+ ANC visits" //Number of months pregnant at time of first ANC visit recode m13 (.=0 "no anc") (0/3=1 "<4") (4 5=2 "4-5") (6 7=3 "6-7") (8/90=4 "8+") (else=9 "don't know/missing"), gen(rh_anc_moprg) label var rh_anc_moprg "Number of months pregnant at time of first ANC visit" //ANC before 4 months recode rh_anc_moprg (0 2/5 9=0 "no") (1=1 "yes"), gen(rh_anc_4mo) lab var rh_anc_4mo "Attended ANC <4 months of pregnancy"</pre> *Place of delivery //Place of delivery recode m15 (20/39 = 1 "Health facility") (10/19 40/98 = 0 "Home/other") (99=9 "Missing"), gen(rh_del_place) label var rh_del_place "Live births by place of delivery" //Assistance during delivery **Note: Assistance during delivery and skilled provider indicators are both country specific indicators. Check the questionnaire and final report for country specific responses for who is a skilled provider. **For Cambodia 2022 FR, doctor, nurse, auxiliary midwife is considered skilled. gen rh_del_pv = 9 replace rh_del_pv = 6 if m3n == 1 = 5 replace rh_del_pv if m3h == 1 | m3k == 1 = 4 replace rh_del_pv if m3g == 1 = 3 replace rh_del_pv if m3c == 1 replace rh_del_pv = 2 if m3b == 1 = 1 replace rh_del_pv if m3a == 1 replace rh_del_pv = 9 if m3a == 8 | m3a == 9 label define pv 111 1 "Doctor" /// 2 "Nurse/midwife" 111 3 "Auxiliary midwife" /// 4 "Traditional birth attendant" 111 5 "Relative/other" /// 111 6 "No one" 9 "Don't know/missing" label val rh_del_pv pv label var rh_del_pv "Person providing assistance during delivery"

//Skilled provider during delivery.

```
** Note: For Cambodia 2022 FR, doctor, nurse, auxiliary midwife is considered skilled.
       recode rh_del_pv (1/3 = 1 "Skilled provider") (4/6 = 2 "Other/No one") (9=3 "Don't
know/missing"), gen(rh_del_pvskill)
       *recode rh_del_pv (1/3 = 1 "Skilled provider") (4/5 = 2 "Unskilled provider") (6 = 3 "No
one") (9=4 "Don't know/missing"), gen(rh_del_pvskill)
       label var rh_del_pvskill "Skilled assistance during delivery"
****Descriptive analysis
foreach var of varlist resid edu maritalstat rc_empl v190 v463a age_b_yr_cat primi-rh_del_pvskill
{
       tab `var', m
}
*Check variables for correlations
pwcorr resid edu maritalstat rc_empl v190 v463a v481 age_b_yr_cat primi-rh_del_pvskill
*v463a dropped due to high level of missing data
*ANC variables also highly correlated. Kept ANC by a skilled provider for analysis.
*Delivery place, provider assistance, and skilled provider all highly correlated. Kept delivery
place for analysis
global stillbirthNRcovs "resid edu maritalstat rc_empl v190 age_b_yr_cat primi histsb
hist_chdeath multpreg rh_anc_pvskill rh_del_place p20"
global istillbirthNRcovs "i.resid i.edu i.maritalstat i.rc_empl i.v190 i.age_b_yr_cat i.primi
i.histsb i.hist_chdeath i.multpreg i.rh_anc_pvskill i.rh_del_place i.p20"
*** Set the complex survey design ***
gen wt=v005/1000000
svyset [pw=wt], psu(v021) strata(v022)
tabout $stillbirthNRcovs stillbirth using NR_sbmodelanalysis_`c'.xls, replace cells(row ci)
cisep(-) f(1 1) svy stats(chi2) pop percent
foreach x in $istillbirthNRcovs {
               svy:logistic stillbirth `x' //Bivariate associations
               outreg2 using "NR_biregs_`c'.xls", eform stats(coef ci) alpha(0.001, 0.01, 0.05)
sideway dec(2) label(insert) noparen nocons append
}
svy: logistic stillbirth i.resid i.edu i.maritalstat i.rc_empl i.v190 i.age_b_yr_cat i.primi
i.hist_chdeath i.multpreg i.rh_anc_pvskill i.rh_del_place i.p20
outreg2 using "NR_multiregs_`c'.xls", eform stats(coef ci) alpha(0.001, 0.01, 0.05) sideway
dec(2) label(insert) noparen nocons append
save "`cn'NR_coded.DTA", replace
}
```